Includes barcode for easy software entry.

HEWLETT-PACKARD

HP-41

USERS' LIBRARY SOLUTIONS

Control Systems



INTRODUCTION

This HP-41C Solutions book was written to help you get the most from your calculator. The programs were chosen to provide useful calculations for many of the common problems encountered.

They will provide you with immediate capabilities in your everyday calculations and you will find them useful as guides to programming techniques for writing your own customized software. The comments on each program listing describe the approach used to reach the solution and help you follow the programmer's logic as you become and expert on your HP calculator.

KEYING A PROGRAM INTO THE HP-41C

There are several things that you should keep in mind while you are keying in programs from the program listings provided in this book. The output from the HP 82143A printer provides a convenient way of listing and an easily understood method of keying in programs without showing every keystroke. This type of output is what appears in this handbook. Once you understand the procedure for keying programs in from the printed listings, you will find this method simple and fast. Here is the procedure:

1. At the end of each program listing is a listing of status information required to properly execute that program. Included is the SIZE allocation required. Before you begin keying in the program, press [XEQ] [ALPHA] SIZE [ALPHA] and specify the allocation (three digits; e.g., 10 should be specified as 010).

   Also included in the status information is the display format and status of flags important to the program. To ensure proper execution, check to see that the display status of the HP-41C is set as specified and check to see that all applicable flags are set or clear as specified.

2. Set the HP-41C to PRGM mode (press the [PRGM] key) and press ■ [GTO] [•] [•] to prepare the calculator for the new program.

3. Begin keying in the program. Following is a list of hints that will help you when you key in your programs from the program listings in this handbook.

   a. When you see " (quote marks) around a character or group of characters in the program listing, those characters are ALPHA. To key them in, simply press [ALPHA], key in the characters, then press [ALPHA] again. So "SAMPLE" would be keyed in as [ALPHA] "SAMPLE" [ALPHA].

   b. The diamond in front of each LBL instruction is only a visual aid to help you locate labels in the program listings. When you key in a program, ignore the diamond.

   c. The printer indication of divide sign is /. When you see / in the program listing, press [÷].

   d. The printer indication of the multiply sign is ⁎. When you see ⁎ in the program listing, press [×].

   e. The ⊢ character in the program listing is an indication of the [APPEND] function. When you see ⊢, press ■ [APPEND] in ALPHA mode (press ■ and the K key).

   f. All operations requiring register addresses accept those addresses in these forms:

      nn (a two-digit number)
      IND nn (INDIRECT: ■ , followed fy a two-digit number)
      X, Y, Z, T, or L (a STACK address: [•] followed by X, Y, Z, T, or L)
      IND X, Y, Z, T or L (INDIRECT stack: ■ [•] followed by X, Y, Z, T, or L)

      Indirect addresses are specified by pressing ■ and then the indirect address. Stack addresses are specified by pressing [•] followed by X, Y, Z, T, or L. Indirect stack addresses are specified by pressing ■ [•] and X, Y, Z, T, or L.

Printer Listing	Keystrokes	Display
01♦LBL "SAMPLE"	■ [LBL] [ALPHA] SAMPLE [ALPHA]	*01 LBL^T SAMPLE*
02 "THIS IS A "	[ALPHA] *THIS IS A* [ALPHA]	*02^T THIS IS A*
03 "⊢SAMPLE"	[ALPHA] ■ [APPEND] SAMPLE	*03^T ⊢ SAMPLE*
04 AVIEW	■ AVIEW [ALPHA]	*04 AVIEW*
05 6	6	*05 6*
06 ENTER↑	[ENTER↑]	*06 ENTER ↑*
07 -2	2 [CHS]	*07 -2*
08 /	[÷]	*08 /*
09 ABS	[XEQ] [ALPHA] ABS [ALPHA]	*09 ABS*
10 STO IND L	[STO] ■ [•] L	*10 STO IND L*
11 "R3="	[ALPHA] R3= ■ [ARCL] 03	*11^T R3=*
12 ARCL 03	■ [AVIEW]	*12 ARCL 03*
13 AVIEW	[ALPHA]	*13 AVIEW*
14 RTN	■ [RTN]	*14 RTN*

TABLE OF CONTENTS

FREQUENCY RESPONSE OF A TRANSFER FUNCTION

For transfer function of the form:

$$G\ (S) = \frac{K_1(Z_2S + 1)}{S^{N_3}(Z_4S + 1)\ (Z_5S + 1)\ (\frac{S^2}{\omega_7^2} + \frac{2Z_6S}{\omega_7} + 1)}$$

The program computes $\angle G\ (j\omega)$, $|G(j\omega)|$ and log $|G(j\omega)|$ for any input frequency ω.

Parameters K_1, Z_2, N_3, Z_4, Z_5, Z_6 and ω_7 are stored in registers 01, 02, 03, 04, 05, 06, and 07 respectively.

NOTE: For type 0 systems, enter $N_3 = 0$. Z_2, Z_4 and/or Z_5 can be entered as 0. If there is no quadratic term, enter Z_6 as 0 and ω_7 very large compared to $\frac{1}{Z_5}$, where Z_5 is the smallest first order term used (other than zero).

<u>Example:</u>

Find $|G(j\omega)|$, $\angle G(j\omega)$ and log $|G(j\omega)|$ for $G(S) = \frac{12(S + 0.6)}{S(S + 1)(S^2 + 6S + 36)}$ frequency 0.01 rad/sec.

First put G(S) into proper form: $G(S) = \frac{.2(1.67S + 1)}{S(S + 1)[(\frac{S}{6})^2 + (\frac{S}{6}) + 1]}$

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 010	
[XEQ] [ALPHA] FREQ [ALPHA]	K1=?
.2 [R/S]	Z2=?
1.67 [R/S]	N3=?
1 [R/S]	Z4=?
1 [R/S]	Z5=?
0 [R/S]	Z6=?
.5 [R/S]	W7=?
6 [R/S]	W=?
.01 [R/S]	G∠JW⅃=20.00
[R/S]	∠G∠JW⅃=-89.71
[R/S]	LOG G=1.30

User Instructions

SIZE: 010

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program			
2	Begin execution, enter data		[XEQ] FREQ	K1=?
		K_1	[R/S]	Z2=?
		Z_2	[R/S]	N3=?
		N_3	[R/S]	Z4=?
		Z_4	[R/S]	Z5=?
		Z_5	[R/S]	Z6=?
		Z_6	[R/S]	W7=?
		ω_7	[R/S]	W=?
3	Answer is automatic	ω	[R/S]	G∠JW⦣=
			[R/S]	∠G∠JW⦣=
			[R/S]	LOG G∠JW⦣=
4	For another frequency		[XEQ] W2	W=?
5	Repeat step 3.			

Program Listings

Store data

```
01◆LBL "FREQ"
02 FIX 2
03 "K1=?"
04 PROMPT
05 STO 01
06 "Z2=?"
07 PROMPT
08 STO 02
09 "N3=?"
10 PROMPT
11 STO 03
12 "Z4=?"
13 PROMPT
14 STO 04
15 "Z5=?"
16 PROMPT
17 STO 05
18 "Z6=?"
19 PROMPT
20 STO 06
21 "W7=?"
22 PROMPT
23 STO 07
24◆LBL "W2"
25 "W=?"
26 PROMPT
27 STO 08
```

$\left[\frac{S^2}{\omega_7^2} + \frac{2Z_6}{\omega_7}\right] + 1$

```
28 RCL 06
29 RCL 07
30 /
31 2
32 *
33 RCL 08
34 *
35 1
36 RCL 08
37 RCL 07
38 /
39 X↑2
40 -
41 R-P
```

$Z_5S + 1$

```
42 RCL 05
43 RCL 08
44 *
45 1
46 R-P
47 XEQ 01
```

$Z_4S + 1$

```
48 RCL 04
49 RCL 08
50 *
51 1
52 R-P
```

S^{N_3}

```
53 XEQ 01
54 RCL 08
55 RCL 03
56 Y↑X
57 *
58 X<>Y
59 RCL 03
60 90
61 *
62 +
63 CHS
64 X<>Y
65 1/X
```

$K_1(Z_2S + 1)$

```
66 RCL 02
67 RCL 08
68 *
69 RCL 01
70 *
71 RCL 01
72 R-P
73 XEQ 01
```

Display

```
74 "G<JW>="
75 ARCL X
76 AVIEW
77 STOP
78 STO 09
79 X<>Y
80 "∡G<JW>="
81 ARCL X
82 AVIEW
83 STOP
84 RCL 09
85 LOG
86 "LOG G="
87 ARCL X
88 AVIEW
89 STOP
```

Multiply complex numbers

```
90◆LBL 01
91 X<>Y
92 RDN
93 *
94 RDN
95 +
96 R↑
97 RTN
98 .END.
```

00

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00		50	
	K_1		
	Z_2		
	N_3		
	Z_4		
05	Z_5	55	
	Z_6		
	ω_7		
	ω		
	$\lvert G(j\omega) \rvert$		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 010 TOT. REG. 34 USER MODE

ENG ___ FIX 2 SCI ___ ON ___ OFF X

DEG X RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

FREQUENCY RESPONSE OF A
TRANSFER FUNCTION
PROGRAM REGISTERS NEEDED: 25

HEWLETT PACKARD
SOLUTION BOOK:
CONTROL SYSTEM

ROW 1 (1 - 3)

ROW 2 (3 - 9)

ROW 3 (9 - 15)

ROW 4 (15 - 20)

ROW 5 (21 - 24)

ROW 6 (25 - 34)

ROW 7 (35 - 47)

ROW 8 (47 - 56)

ROW 9 (57 - 68)

ROW 10 (69 - 74)

ROW 11 (74 - 80)

ROW 12 (80 - 86)

ROW 13 (86 - 96)

ROW 14 (97 - 98)

BODE OF TRANSFER FUNCTION THAT HAS EACH POLE AND ZERO GIVEN

Given a laplace transfer function:

$$F(s) = K \frac{(s + z_1 + jz_1')\ (s + z_2 + jz_2')\ \ldots\ (s + z_N + jz_N')}{(s + p_1 + jp_1')\ (s + p_2 + jp_2')\ \ldots\ (s + z_M + jz_M')}$$

F(jw) = Magnitude $\angle\ \theta$

$$\text{where magnitude} = |K| \frac{\sqrt{(z_1' + \omega)^2 + z_1^2}\ \sqrt{(z_2' + \omega)^2 + z_2^2}\ \ldots \sqrt{(z_N' + \omega)^2 + z_N^2}}{\sqrt{(p_1' + \omega)^2 + p_1^2}\ \sqrt{(p_2' + \omega)^2 + p_2^2}\ \ldots \sqrt{(p_M' + \omega)^2 + p_M^2}}$$

$$\text{and } \theta = \text{Tan}^{-1}\left(\frac{z_1' + \omega}{z_1}\right) + \text{Tan}^{-1}\left(\frac{z_2' + \omega}{z_2}\right) + \ldots + \text{Tan}^{-1}\left(\frac{z_N' + \omega}{z_N}\right) - \text{Tan}^{-1}\left(\frac{p_1' + \omega}{p_1}\right)$$

$$+ \text{Tan}^{-1}\left(\frac{p_2' + \omega}{p_2}\right) + \ldots + \text{Tan}^{-1}\left(\frac{p_M' + \omega}{p_M}\right) + 2\ \frac{|K| - K}{K}\ \text{Tan}^{-1} 1$$

and decibels = 20 $\log_{10}$ magnitude.

N.B. $\text{zero}_i = -(z_i + jz_i')$

$\text{pole}_i = -(p_i + jp_i')$

NOTE: K must be real and non-zero but may be negative.

Example:

Find the gain (in decibels and volts/volt) and phase shift in degrees for:

$$F(s) = 0.09091\left(\frac{s + 62890}{s + 5717}\right)$$

at 3,333 Hz.

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 008	
[XEQ] [ALPHA] BODE1 [ALPHA]	FREQ?
3333 [R/S]	N?
1 [R/S]	Z?
62890 [R/S]	ZP?
0 [R/S]	M?
1 [R/S]	P?
5717 [R/S]	PP?
0 [R/S]	K?
.09091 [R/S]	dB=-11.1
[R/S]	MAG=2.7759E-1
[R/S]	∡=-56.3

User Instructions

SIZE: 008

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program.			
2	Begin execution:		[XEQ] BODE1	FREQ?
	input frequency in Hertz	f	[R/S]	N?
	input number of highest zero	N	[R/S]	Z?
	input Z_1	Z_1	[R/S]	ZP?
	input Z_1'	Z_1'	[R/S]	Z?
	input Z_2	Z_2	[R/S]	ZP?
	input Z_2'	Z_2'	[R/S]	Z?
	⋮	⋮	⋮	⋮
	continue until Z_N and Z_N' are in.			
				M?
	input number of highest pole	M	[R/S]	P?
	input P_1	P_1	[R/S]	PP?
	input P_1'	P_1'	[R/S]	P?
	⋮	⋮	⋮	⋮
	continue until P_N and P_N' are in.			
				K?
	input K.	K	[R/S]	
3	Solution - decibels			dB=
	- magnitude		[R/S]	MAG=
	- phase angle		[R/S]	∡ =

Program Listings

Initialization

```
01♦LBL "BOD
E1"
02 CLRG
03 "FREQ.?"
04 PROMPT
05 STO 06
06 2
07 *
08 PI
09 *
10 STO 05
11 1
12 STO 01
13 "N?"
14 PROMPT
15 1
16 -
17 1000
18 /
19 STO 00
20♦LBL B
21 "Z?"
22 PROMPT
23 "ZP?"
24 PROMPT
25 RCL 03
26 1
27 +
28 STO 03
29 STO 07
30 RDN
31 RCL 05
32 +
33 X<>Y
34 R-P
35 XEQ 00
36 ISG 00
37 GTO B
38 "M?"
39 PROMPT
40 1
41 -
42 1000
43 /
44 STO 00
45♦LBL C
46 "P?"
47 PROMPT
48 "PP?"
49 PROMPT
```

```
50 RCL 04
51 1
52 -
53 STO 04
54 STO 07
55 RDN
56 RCL 05
57 +
58 CHS
59 X<>Y
60 R-P
61 1/X
62 XEQ 00
63 ISG 00
64 GTO C
65 0
66 "K?"
67 PROMPT
68 R-P
69 XEQ 00
70 RCL 01
71 LOG
72 20
73 *
```

Output

```
74 FIX 1
75 "dB="
76 XEQ d
77 RCL 01
78 SCI 4
79 "MAG="
80 XEQ d
81 RCL 02
82 FIX 1
83 "∡="
84♦LBL d
85 ARCL X
86 AVIEW
87 STOP
88 RTN
```

Subroutine to update magnitude

```
89♦LBL 00
90 ST* 01
91 RDN
92 RCL 02
93 +
94 1
95 P-R
96 R-P
97 RDN
98 STO 02
99 RTN
100 .END.
```

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	N, M max	50	
	Magnitude M		
	Phase Ang. θ		
	Zero counts i < N		
	pole counts -1>-M		
05	radians/second	55	
	first input freq		
	last updated norm		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 008	TOT. REG. 32	USER MODE
ENG ___	FIX ___ SCI ___	ON ___ OFF X
DEG x	RAD ___ GRAD ___	

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

BODE OF TRANSFER FUNCTION THAT
HAS EACH POLE AND ZERO GIVEN
PROGRAM REGISTERS NEEDED: 25

HEWLETT PACKARD
SOLUTION BOOK:
CONTROL SYSTEM

ROW 1 (1 – 3)

ROW 2 (3 – 12)

ROW 3 (13 – 20)

ROW 4 (20 – 27)

ROW 5 (28 – 37)

ROW 6 (37 – 43)

ROW 7 (44 – 50)

ROW 8 (51 – 62)

ROW 9 (62 – 69)

ROW 10 (69 – 75)

ROW 11 (76 – 80)

ROW 12 (80 – 87)

ROW 13 (88 – 99)

ROW 14 (100 – 100)

FREQUENCY RESPONSE OF A THIRD OVER A FOURTH ORDER TRANSFER

by Craig A. Lincoln

Given a transfer function of the form:

$$F(s) = \frac{(N3)s^3 + (N2)s^2 + (N1)s + (N0)}{(D4)s^4 + (D3)s^3 + (D2)s^2 + (D1)s + (D0)}$$

$$\text{the magnitude} = \sqrt{\frac{(N0 - N2\ \omega^2)^2 + (N1\ \omega - N3\ \omega^3)^2}{(D0 - D2\ \omega^2 + D4\ \omega^4)^2 + (D1\ \omega - D3\ \omega^3)^2}}$$

where $\omega = 2\pi f$ and magnitude in dB = 20 $\log_{10}$ magnitude

$$\text{The phase angle} = \text{ATAN}\left(\frac{N1\ \omega - N3\ \omega^3}{N0 - N2\ \omega^2}\right) - \text{ATAN}\left(\frac{D1\ \omega - D3\ \omega^3}{D0 - D2\ \omega^2 + D4\ \omega^4}\right)$$

Phase angle in degrees unless angular mode is changed.

NOTE: N3 and D4 must be -1, 0 or +1.
All coefficient values must be real.
Low frequency must be $> 10^{-40}$.
Multiplier must be > 1.

References: 1. Kuo, B. C., Automatic Control Systems, Prentice-Hall, 1982.
2. University of Illinois, E.E. Curriculum, Class Notes.

EXAMPLE:

Find the gain (in decibels and volts/volt) and phase shift (in degrees) for:

$$F(s) = \frac{8s^3 - 1}{3s^3 + 8}$$

at 0.01, 0.1 and 1.0 Hz

Rewriting the function yields:

$$F(s) = \frac{S^3 - 0.125}{0.375\ S^3 + 1}$$

N3 = 1.0	D4 = 0	D0 = 1
N2 = 0	D3 = 0.375	Low Freq. = 0.01
N1 = 0	D2 = 0	High Freq. = 1.0
N0 = -0.125	D1 = 0	Multiplier = 10

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 012	
[XEQ] [ALPHA] FR2 [ALPHA]	HZ OR RAD?
HZ [R/S]	dB? <Y,N>
Y [R/S]	MAG? <Y,N>
Y [R/S]	∡? <Y,N>
Y [R/S]	N3?
1 [R/S]	N2?
0 [R/S]	N1?
0 [R/S]	N0?
-.125 [R/S]	D4?
0 [R/S]	D3?
.375 [R/S]	D2?
0 [R/S]	D1?
0 [R/S]	D0?
1 [R/S]	LOW FREQ?
.01 [R/S]	MULTIPLIER
10 [R/S]	HIGH FREQ?
1 [R/S]	FREQ=0.0100
[R/S]*	dB=-18.0618
[R/S]*	MAG=0.1250
[R/S]*	∡=-179.8810

Keystrokes:	Display:
[R/S]*	FREQ=0.1000
[R/S]*	dB=-11.1638
[R/S]*	MAG=0.2766
[R/S]*	∢=-111.4306
[R/S]*	FREQ=1.000
[R/S]*	dB=8.5189
[R/S]*	MAG=2.6665
[R/S]*	∢=-0.6448

For the given transfer function, find the magnitude, dB, and phase angle at 0.0 radians (assumes the above example has been run).

Note: Use 10^{-40} instead of zero. 0.0 radians = 0.0 Hertz

Keystrokes:	Display:
[XEQ] [ALPHA] RG [ALPHA]	LOW FREQ?
[EEX] [CHS] 40	MULTIPLIER
2 [R/S]	HIGH FREQ?
0 [R/S]	FREQ=1.0000E-40
[R/S]*	dB=-18.0618
[R/S]*	MAG=0.1250
[R/S]*	∢=180.0000

User Instructions

SIZE: 012

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program			
2	Begin execution		[XEQ] "FR2"	
	If SIZE too small			XEQ SIZE 012
	Reset to 012		[XEQ] "SIZE" 012	
	Continue		[R/S]	HZ OR RAD?
3	Enter units	HZ or RAD	[R/S]	dB? <Y,N>
4	Output in dB	Y or N	[R/S]	MAG? <Y,N>
5	Magnitude Output	Y or N	[R/S]	∢? <Y,N>
6	Phase Angle Output	Y or N	[R/S]	N3?
7	Enter Numerator Coefficient	N3	[R/S]	N2?
8	" " "	N2	[R/S]	N1?
9	" " "	N1	[R/S]	N0?
10	" " "	N0	[R/S]	D4?
11	Enter Denominator Coefficient	D4	[R/S]	D3?
12	" " "	D3	[R/S]	D2?
13	" " "	D2	[R/S]	D1?
14	" " "	D1	[R/S]	D0?
15	" " "	D0	[R/S]	LOW FREQ?
16	Enter lowest frequency	Low	[R/S]	MULTIPLIER
17	Enter multiplier	Mult	[R/S]	HIGH FREQ?
18	Enter high frequency	High	[R/S]*	
	NOTE: For the response to a single fre-			
	quency, enter that frequency as low, 2 as			
	the multiplier, and 0 for high.			
19	To enter a new range		[XEQ] "RG"	
	Repeat steps 16-19.			

*This keystroke not necessary if printer is in use.

Program Listings

Program	Comments
`01♦LBL "FR2"`	
`02 SF 21`	Initialization
`03 SF 25`	
`04 RCL 11` `05 "XEQ SIZE 012"`	Check for correct size
`06 FC?C 25`	
`07 PROMPT`	
`08 DEG`	
`09 FIX 4`	Set display mode

`10 SF 00`
`11 CF 01`
`12 SF 02`
`13 CF 03`
`14 CF 04`
`15 CF 05`
`16 CF 06`
`17 CF 07`
	Initialize all user-defined flags

`18 AON`	
`19 "HZ OR RAD?"` `20 PROMPT`	Prompt for frequency units

`21 AOFF`
`22 ASTO X`
`23 "RAD"`
`24 ASTO Y`
`25 X=Y?`	

`26 SF 04` `27 "dB?"`	Set flag 04 if rad/sec
`28 XEQ 04`	
`29 X=Y?` `30 SF 05`	Prompt for desired output

`31 "MAG?"`
`32 XEQ 04`
`33 X=Y?`
`34 SF 06`
`35 "∡?"`
`36 XEQ 04`
`37 X=Y?`
`38 SF 07`
`39 "N3?"`
	Set the appropriate flags

`40 PROMPT`
`41 X=0?`
`42 CF 00`
`43 X<0?`
`44 SF 01`
`45 "N2?"`
`46 PROMPT`
`47 STO 01`
`48 "N1?"`
`49 PROMPT`
`50 STO 02`
`51 "N0?"`
`52 PROMPT`
`53 STO 03`
`54 "D4?"`
	Prompt for each of the numerator coefficients

`55 PROMPT`
`56 X=0?`
`57 CF 02`
`58 X<0?`
`59 SF 03`
`60 "D3?"`
`61 PROMPT`
`62 STO 04`
`63 "D2?"`
`64 PROMPT`
`65 STO 05`
`66 "D1?"`
`67 PROMPT`
`68 STO 06`
`69 "D0?"`
`70 PROMPT`
`71 STO 07`
	Prompt for each of the denominator coefficients

`72♦LBL "RG"`
`73 "LOW FREQ?"`
`74 PROMPT`
`75 X=0?`
`76 GTO "RG"`
`77 STO 10`
`78 1`
`79 "MULTIPLIER"`
`80 PROMPT`
`81 X<=Y?`
`82 GTO "RG"`
`83 STO 09`
`84 "HIGH FREQ?"`
`85 PROMPT`
`86 STO 11`
`87 GTO 01`
	Prompt for frequency range and step function value

`88♦LBL 16`
`89 RCL 11`
`90 RCL 10`
`91 RCL 09`
`92 *`
`93 STO 10`
`94 X>Y?`
`95 STOP`
	Compute the next frequency to be used (used for calculating results for range of frequency).

`96♦LBL 01`	

Program Listings

Program	Description
ADV	
ADV	
RCL 10	Display the current frequency
"FREQ="	
ARCL X	
AVIEW	
FS? 04	
GTO 00	Convert units to rad/sec if input was given in Hertz
2	
*	
PI	
*	
♦LBL 00	
STO 08	
RCL 02	Calculate linear magnitude
X<>Y	
/	
FS? 00	
RCL 08	
FS? 01	(Equation given in Program Description)
CHS	
FS? 00	
-	
RCL 03	
RCL 08	
X↑2	
/	
RCL 01	
-	
R-P	
STO 00	
RDN	
RCL 06	
RCL 08	
/	
RCL 04	
RCL 08	
*	
-	
RCL 07	
RCL 08	
X↑2	
/	
FS? 02	
LASTX	
FS? 03	
CHS	
FS? 02	
+	
RCL 05	
-	

Program	Description
RCL 00	
RDN	
R-P	
RDN	
RDN	
STO 00	
RDN	
X<>Y	
/	
ENTER↑	
LOG	Magnitude in dB
20	
*	
"dB="	
FS? 05	Display magnitude in decibels
XEQ 03	
X<>Y	
"MAG="	
FS? 06	Display linear magnitude
XEQ 03	
FC? 07	
GTO 16	
R↑	Calculate and display the phase angle
R↑	
RCL 00	
-	
CHS	
LASTX	
X<>Y	
1	
P-R	
R-P	
RDN	
"∡="	
XEQ 03	
GTO 16	
♦LBL 03	Display subroutine
ARCL X	
AVIEW	
RTN	
♦LBL 04	Subroutine used in prompting for desired output
AON	
"⊢<Y,N>"	
PROMPT	
AOFF	
ASTO X	
"Y"	
ASTO Y	
END	

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	Used for temp. storage in calc. mag.
01	N2
02	N1
03	N0
04	D3
05	D2
06	D1
07	D0
08	Current Frequency
09	Multiplier
10	Lowest Frequency
11	Highest Frequency

STATUS

SIZE 012 TOT. REG. 67 USER MODE
ENG FIX 4 SCI ON OFF
DEG X RAD GRAD

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
00	S	N3 ≠ 0	N3 = 0
01	C	N3 < 0	N3 ≥ 0
02	S	D4 ≠ 0	D4 = 0
03	C	D4 < 0	D4 ≥ 0
04	C	RAD/SEC	HERTZ
05	C	dB output on	dB output off
06	C	Mag output on	Mag output off
07	C	∡ output on	∡ output off
21	S	Printer enable	Printer disable
25	S	Error ignore on	Error ignore off

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

FREQ RESPONSE OF A THIRD OVER
A FOURTH ORDER TRANSFER
PROGRAM REGISTERS NEEDED: 55

HEWLETT PACKARD
SOLUTIONS BOOK:
CONTROL SYSTEMS

ROW 1 (1 : 5)

ROW 2 (5 : 6)

ROW 3 (6 : 13)

ROW 4 (14 : 19)

ROW 5 (19 : 23)

ROW 6 (23 : 28)

ROW 7 (28 : 33)

ROW 8 (34 : 39)

ROW 9 (39 : 45)

ROW 10 (46 : 52)

ROW 11 (53 : 60)

ROW 12 (60 : 66)

ROW 13 (66 : 72)

ROW 14 (72 : 73)

ROW 15 (74 : 79)

ROW 16 (79 : 83)

ROW 17 (84 : 86)

ROW 18 (87 : 97)

FREQ RESPONSE OF A THIRD OVER
A FOURTH ORDER TRANSFER

HEWLETT PACKARD
SOLUTIONS BOOK:
CONTROL SYSTEMS

ROW 19 (98 : 103)

ROW 20 (104 : 114)

ROW 21 (115 : 125)

ROW 22 (126 : 138)

ROW 23 (139 : 148)

ROW 24 (149 : 160)

ROW 25 (161 : 165)

ROW 26 (165 : 170)

ROW 27 (171 : 181)

ROW 28 (182 : 189)

ROW 29 (190 : 194)

ROW 30 (195 : 196)

BODE OF THIRD-ORDER OVER THIRD-ORDER TIMES S^N TRANSFER FUNCTION

$$F(s) = \frac{Qs^3 + Gs^2 + Hs + K}{s^N(Rs^3 + Ls^2 + Ms + P)}$$

where: Q and R are options, limited to [-1, 0, +1], G, H, K, L, M, and P may assume any real value, and N is a non-negative integer [0, 1, 2, ...].

$$\text{Magnitude} = \frac{\sqrt{(K - G\omega^2)^2 + (H\omega - Q^3)^2}}{\omega N \sqrt{(P - L\omega^2)^2 + (M\omega - R^3)^2}}$$

where: $\omega = 2\pi F$

$$\text{Decibels} = 20 \log_{10} \text{magnitude}$$

$$\text{Phase} = \text{Tan}^{-1}\left(\frac{H\omega - Q\omega^3}{K - G\omega^2}\right) - N\sin^{-1} 1.0 - \text{Tan}^{-1}\left(\frac{M\omega - R\omega^3}{P - L\omega^2}\right)$$

in degrees (unless angular mode is set to radians or grads).

NOTE: Frequency, f, cannot equal zero. Power, N, is a non-negative integer. Coefficient of s^3 in numerator or denominator is -1, 0, or +1. Display of phase will be between -180° and +180° ($-\pi$ and $+\pi$ radians or -200 and +200 grads).

Example:

Find gain (in decibels and volts/volt) and phase shift in degrees for:

$$F(s) = \frac{8s^3 - 1}{s^2(s^2 - 2)}$$

at f = 0.1 Hz and 1.0 Hz.

Rewritten: $$F(s) = \frac{s^3 - 0.125}{s^2(0.125s^2 - 0.25)}$$

Q = 1	G = 0
H = 0	K = -0.125
N = 2	R = 0
L = 0.125	M = 0
P = -0.25	f = 0.1 and 1.0

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 010	
[XEQ] [ALPHA] BODE3 [ALPHA]	Q?
1 [R/S]	G?
0 [R/S]	H?
0 [R/S]	K?
.125 [CHS] [R/S]	N?
2 [R/S]	R?
0 [R/S]	L?
.125 [R/S]	M?
0 [R/S]	P?
.25 [CHS] [R/S]	FREQ.?
.1 [R/S]	dB=7.4229
[R/S]	MAG=2.3504
[R/S]	∡=-116.7449
[R/S]	FREQ.?
1 [R/S]	dB=1.6690
[R/S]	MAG=1.2118
[R/S]	∡=-90.0289

User Instructions

SIZE: 010

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program.			
2	Begin execution and input constants		[XEQ] BODE3	Q?
	(remember, Q and R must have values +1,	Q	[R/S]	G?
	0, or -1).	G	[R/S]	H?
		H	[R/S]	K?
		K	[R/S]	N?
		N	[R/S]	R?
		R	[R/S]	L?
		L	[R/S]	M?
		M	[R/S]	P?
		P	[R/S]	FREQ.?
3	Input frequency (Hertz).	f	[R/S]	
4	Solution - decibels			dB=
	- magnitude		[R/S]	MAG=
	- phase angle		[R/S]	∡ =
5	For a new frequency.		[R/S]	FREQ.?
6	Go to step 3.			

Program Listings

Initialization

```
01♦LBL "BOD
E3"
02 SF 00
03 CF 01
04 SF 02
05 CF 03
06 "Q?"
07 PROMPT
08 X=0?
09 CF 00
10 X<0?
11 SF 01
12 "G?"
13 PROMPT
14 STO 01
15 "H?"
16 PROMPT
17 STO 02
18 "K?"
19 PROMPT
20 STO 03
21 "N?"
22 PROMPT
23 STO 04
24 "R?"
25 PROMPT
26 X=0?
27 CF 02
28 X<0?
29 SF 03
30 "L?"
31 PROMPT
32 STO 05
33 "M?"
34 PROMPT
35 STO 06
36 "P?"
37 PROMPT
38 STO 07
39♦LBL A
40 "FREQ.?"
41 PROMPT
```

Convert to radians

```
42 2
43 *
44 PI
45 *
46 STO 08
47 RCL 06
48 FC? 02
49 GTO 02
50 RCL 08
51 X↑2
52 FS? 03
53 CHS
54 -
55♦LBL 02
56 RCL 07
57 RCL 08
58 /
59 RCL 05
60 RCL 08
61 *
62 -
63 R-P
64 RCL 08
65 RCL 04
66 Y↑X
67 *
68 1
```

Denominator magnitude

```
69 ASIN
70 RCL 04
71 *
72 X<>Y
73 RDN
74 +
75 R↑
76 STO 09
77 RDN
78 RCL 03
79 RCL 08
80 /
81 RCL 01
82 RCL 08
83 *
84 -
85 RCL 02
86 FC? 00
87 GTO 01
88 RCL 08
89 X↑2
90 FS? 01
91 CHS
92 -
93♦LBL 01
94 X<>Y
95 RCL 09
96 RDN
97 R-P
98 R↑
99 /
100 RDN
101 -
```

Program Listings

```
102 CHS
103 R↑
104 ENTER↑
105 LOG
106 20
107 *
108 "dB="
109 XEQ d
110 X<>Y
111 "MAG="
112 XEQ d
113 R↑
114 1
115 P-R
116 R-P
117 RDN
118 "∡="
119 XEQ d
120 GTO A
121♦LBL d
122 ARCL X
123 AVIEW
124 STOP
125 RTN
126 .END.
```

Display

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00		50	
	G		
	H		
	K		
	N		
05	L	55	
	M		
	P		
	W		
	Temp storage		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 010 TOT. REG. 37 USER MODE
ENG ___ FIX 4 SCI ___ ON ___ OFF X
DEG ___ RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
00		$Q \neq 0$	$Q = 0$
01		$Q < 0$	$Q \geq 0$
02		$R \neq 0$	$R = 0$
03		$R < 0$	$R \geq 0$

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

BODE OF 3RD–ORDER/3RD–ORDER *
S^N TRANSFER FUNCTION
PROGRAM REGISTERS NEEDED: 28

HEWLETT PACKARD
SOLUTION BOOK:
CONTROL SYSTEM

ROW 1 (1 – 3)

ROW 2 (4 – 11)

ROW 3 (11 – 18)

ROW 4 (18 – 26)

ROW 5 (27 – 33)

ROW 6 (34 – 40)

ROW 7 (40 – 49)

ROW 8 (49 – 60)

ROW 9 (61 – 73)

ROW 10 (74 – 86)

ROW 11 (86 – 96)

ROW 12 (97 – 108)

ROW 13 (108 – 112)

ROW 14 (112 – 119)

ROW 15 (120 – 126)

ROUTH TEST FOR CONTINUOUS AND DISCRETE TIME SYSTEM STABILITY

This program is designed to test the location of the roots of a characteristic equation in the form:

$$A_0 S^n + A_1 S^{n-1} + A_2 S^{n-2} \; . \; . \; . \; A_{n-1} S + An = 0$$

using the repeated test:

$$K = A_0/A_1 \quad \text{for } i = 1,\, 3,\, 5 \; . \; . \; . \quad \begin{cases} A_{i-1} = A \\ A_i = A_{i+1} - KA_{i-2} \end{cases}$$

The number of right hand plane (RHP) roots is equal to Σ(# times $K < 0$).

NOTES: A must be $\leqslant$ 20.

Routh test with various roots along imaginary axis may indicate instability. Shift axis slightly to check actual location of roots.

Axis shift:

$$\begin{bmatrix} A_1 = A_1 + A_0 d \\ A_2 = A_2 + A_1 d \\ \vdots \quad \vdots \quad \vdots \\ A_n = A_n + A_{n-1} d \end{bmatrix} \quad \begin{bmatrix} A_1 = A_1 + A_0 d \\ A_2 = A_2 + A_1 d \\ \vdots \quad \vdots \quad \vdots \\ A_{n-1} = A_{n-1} + A_{n-2} d \end{bmatrix} \begin{bmatrix} A_1 = A_1 + A_0 d \\ A_2 = A_2 + A_1 d \end{bmatrix} \quad [A_1 = A_1 + A_0 d]$$

Example:

Does $S^4 + 2s^3 + S^2 + 4S + 4 = 0$ represent a stable system?

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 026	
[XEQ] [ALPHA] ROUTH [ALPHA]	N?
4 [R/S]	A0?
1 [R/S]	A1?
2 [R/S]	A2?
1 [R/S]	A3?
4 [R/S]	A4?
4 [R/S]	TEST? SHIFT?
[XEQ] [ALPHA] TEST [ALPHA]	1.000 00
	2.000 00
	-1.000 00
	1.200 01
	4.000 00
	RHP ROOTS 2

Because the RHP roots is 2, the system is unstable.

User Instructions

SIZE: 026

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program.			
2	Begin execution and enter equation.		[XEQ] ROUTH	N?
		n	[R/S]	A0?
		A_0	[R/S]	A1?
		A_1	[R/S]	A2?
		A_2	[R/S]	⋮
		⋮	[R/S]	A(n)?
		A_n	[R/S]	TEST? SHIFT?
3	For Routh test:		[XEQ] TEST	A0
	first column of routh array is displayed			⋮
	using PSE, stopping to display the #RHP			RHP ROOT
	roots.			
4	Axis Shift: the coefficients to relocate			
	equation to reflect an axis shift		[XEQ] SHIFT	d?
	input d	d	[R/S]	A0
	new coefficients are displayed using PSE.			A1
				⋮
				A(n)

Program Listings

Initialization

```
 01♦LBL "ROU
TH"
 02 "N?"
 03 PROMPT
 04 STO 25
 05 0
 06 STO 24
 07 CF 29
 08 FIX 0
 09♦LBL e
 10 "A"
 11 ARCL 24
 12 "⊢?"
 13 PROMPT
 14 STO IND
24
 15 RCL 25
 16 RCL 24
 17 X=Y?
 18 GTO A
 19 1
 20 ST+ 24
 21 GTO e
 22♦LBL A
 23 "TEST? S
HIFT?"
 24 PROMPT
```

Routh test

```
 25♦LBL "TES
T"
 26 SCI 3
 27 0
 28 STO 23
 29♦LBL 00
 30 RCL 00
 31 PSE
 32 RCL 01
 33 X≠0?
 34 GTO a
 35 CLX
 36  E-50
 37 STO 01
 38♦LBL a
 39 /
 40 STO 22
 41 X>0?
 42 GTO a
 43 RCL 23
 44 1
 45 +
 46 STO 23
 47♦LBL a
 48 1
 49 STO 25
 50♦LBL 01
 51 RCL IND
25
 52 DSE 25
 53 STO IND
25
 54 STO IND
25
 55 2
 56 ST+ 25
 57 RCL IND
25
 58 ISG 25
 59♦LBL 50
 60 RCL IND
25
 61 RCL 22
 62 *
 63 -
 64 DSE 25
 65 DSE 25
 66 STO IND
25
 67 2
 68 ST+ 25
 69 RCL 25
 70 RCL 24
 71 X>Y?
 72 GTO 01
 73 1
 74 -
 75 STO 24
 76 RCL IND
25
 77 0
 78 STO IND
25
 79 DSE 25
 80 RDN
 81 STO IND
25
 82 RCL 24
 83 X>0?
 84 GTO 00
 85 RCL 00
 86 PSE
 87 "RHP ROO
TS "
 88 FIX 0
```

Program Listings

```
 89 ARCL 23
 90 AVIEW
 91 STOP
 92♦LBL "SHI
FT"                          Axis shift
 93 "d?"
 94 PROMPT
 95 SCI 4
 96 STO 23
 97 RCL 24
 98 STO 22
 99♦LBL 02
100 0
101 STO 25
102♦LBL 03
103 RCL IND
25
104 RCL 23
105 *
106 ISG 25
107 "A"
108 ST+ IND
25
109 RCL 25
110 RCL 22
111 X≠Y?
112 GTO 03
113 1
114 -
115 STO 22
116 X≠0?
117 GTO 02
118 SCI 5
119 0
120 STO 25
121♦LBL 08
122 RCL IND
25
123 STOP                     Display
124 ISG 25
125 "A"
126 RCL 24
127 RCL 25
128 X<=Y?
129 GTO 08
130 .END.
```

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	A_0	50	
	A_1		
	A_2		
	·		
	·		
05	·	55	
	·		
	·		
	·		
	·		
10	·	60	
	·		
	·		
	·		
	·		
15	·	65	
	·		
	·		
	·		
	·		
20	A_{19}	70	
	A_{20}		
	temp. storage		
	temp. storage		
	temp. storage		
25	n	75	
	reserved		
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 026 TOT. REG. 64 USER MODE
ENG ____ FIX ____ SCI ____ ON ____ OFF X
DEG X RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

ROUTH TEST FOR CONTINUOUS AND DISCRETE TIME SYSTEM STABILITY
PROGRAM REGISTERS NEEDED: 38

HEWLETT PACKARD
SOLUTION BOOK:
CONTROL SYSTEM

ROW 1 (1 - 3)

ROW 2 (4 - 10)

ROW 3 (11 - 17)

ROW 4 (18 - 23)

ROW 5 (23 - 25)

ROW 6 (25 - 29)

ROW 7 (30 - 37)

ROW 8 (38 - 45)

ROW 9 (46 - 53)

ROW 10 (53 - 60)

ROW 11 (60 - 68)

ROW 12 (68 - 76)

ROW 13 (76 - 84)

ROW 14 (84 - 87)

ROW 15 (87 - 92)

ROW 16 (92 - 97)

ROW 17 (98 - 106)

ROW 18 (106 - 113)

ROW 19 (114 - 122)

ROW 20 (122 - 129)

ROW 21 (130 - 130)

CONVERT FREQUENCY RESPONSE -- OPEN LOOP, CLOSED LOOP

For a linear, unity feedback control system this program converts open loop frequency response data $\measuredangle G(j\omega)$ (or log $|G(j\omega)|$ or 20 log $|G(j\omega)|$) to closed loop data $\measuredangle C/R(j\omega)$ and $|C/R(j\omega)|$ (or log $C/R(j\omega)$ or 20 log - $(j\omega)$).



where $\frac{C}{R}(j\omega) = \frac{G(j\omega)}{1 + G(j\omega)}$

This program also converts from closed loop data to open loop data.

$\measuredangle \frac{C}{R}(j\omega)$ and $\left|\frac{C}{R}(j\omega)\right|$
$\measuredangle G(j\omega)$ and $|G(j\omega)|$ } the relationship used is: $g(j\omega) = \dfrac{\frac{C}{R}(j\omega)}{1 - \frac{C}{R}(j\omega)}$

Note:

A. When the input phase angle cosine ($\measuredangle C/R(j\omega)$ or $\measuredangle G(j\omega)$) is zero (e.g., $\measuredangle C/R(j\omega) = -90$), the conversion is only accurate through the 6th decimal place.

Example:

For the system: $G(j\omega) = \dfrac{4}{j\omega(1 + .25j\omega)\ (1 + .0625j\omega)}$

For ω = .1, 1 and 10 rad/sec, ∡G(jω) and |G(jω)| are:

ω rad/sec	∡G(jω) degrees	|G(jω)|
.1	-91.79	39.99
1	-107.61	3.87
10	-190.20	.13

Determine closed loop frequency response for ω = .1, 1, 10 rad/sec.

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 007	
[XEQ] [ALPHA] O-C [ALPHA]	∡G?
91.79 [CHS] [R/S]	C?
39.99 [R/S]	C/R=1.00
[R/S]	∡C/R=-1.43
[XEQ] [ALPHA] O-C [ALPHA]	∡G?
107.61 [CHS] [R/S]	G?
3.87 [R/S]	C/R=1.05
[R/S]	∡C/R=-14.96
[XEQ] [ALPHA] O-C [ALPHA]	∡G?
190.2 [CHS] [R/S]	G?
.13 [R/S]	C/R=0.15
[R/S]	∡G/R=-191.71

User Instructions

SIZE: 007

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program.			
2	To convert from open loop to closed loop			
	response,		[XEQ] O - C	∡ G?
	input angle of G(jω)	∡ G(jω)	[R/S]	G?
	input magnitude G(jω)	G(jω)	[R/S]	C/R=
			[R/S]	∡ C/R=
3	To convert from closed loop to open loop			
	response,		[XEQ] C - O	∡ C/R?
	input angle C/R(jω)	∡C/R(jω)	[R/S]	C/R?
	input magnirude C/R(jω)	C/R(jω)	[R/S]	G=
			[R/S]	∡ G=
4	For a new case		[R/S]	
	go to step 2 or 3.			

Program Listings

Open to Closed

```
01♦LBL "O-C"
02 SF 03
03 0
04 STO 06
05♦LBL 05
06 "∡C/R?"
07 FS? 03
08 "∡G?"
09 PROMPT
10 STO 04
11 RCL 06
12 -
13 1
14 "C/R?"
15 FS? 03
16 "G?"
17 PROMPT
18 GTO C
```

Closed to Open

```
19♦LBL "C-O"
20 CF 03
21 180
22 STO 06
23 GTO 05
```

Complex Algebra

```
24♦LBL C
25 X<>Y
26 RDN
27 STO 05
28 P-R
29 1
30 +
31 R-P
32 RCL 05
33 X<>Y
34 /
35 STO 01
36 X<>Y
37 CHS
38 RCL 04
39 +
40 STO 02
41 X<>Y
42 "G="
43 FS? 03
44 "C/R="
45 XEQ d
46 RCL 02
47 "∡G="
48 FS? 03
49 "∡C/R="
50 XEQ d
51 FS? 03
52 GTO "O-C"
53 GTO "C-O"
```

Display

```
54♦LBL d
55 FIX 2
56 ARCL X
57 AVIEW
58 STOP
59 RTN
60 .END.
```

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00		50	
	comp. mag.		
	comp. plan		
	entered mag.		
05	entered phase	55	
	0 or 180		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 007 TOT. REG. 25 USER MODE
ENG ___ FIX 2 SCI ___ ON ___ OFF X
DEG X RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
03		0 - C	C - 0

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

CONVERT FREQUENCY RESPONSE
-- OPEN LOOP & CLOSED LOOP
PROGRAM REGISTERS NEEDED: 19

HEWLETT PACKARD
SOLUTION BOOK:
CONTROL SYSTEM

ROW 1 (1 - 6)

ROW 2 (6 - 10)

ROW 3 (11 - 16)

ROW 4 (17 - 20)

ROW 5 (21 - 29)

ROW 6 (30 - 42)

ROW 7 (42 - 46)

ROW 8 (47 - 50)

ROW 9 (50 - 53)

ROW 10 (53 - 60)

AID TO ROOT LOCUS PLOTS I - REAL POLES

Given the forward transfer function of unity gain feedback system:

$$KG(s) = \frac{K(s + z_1)\ (s + z_2)}{s(s + p_1)\ (s + p_2)\ (s + p_3)\ (s + p_4)}$$

where $s = \sigma + i\omega$ is the complex frequency variables and z_1, z_2, p_1, p_2, p_3 and p_4 are real numbers, the program helps in finding the roots of $1 + KG(s) = 0$, which determines the poles of the closed-loop system. It follows that at any point in the s-plane, which is a root of the above equation for some value of K, $G(\sigma + i\omega) = 1/K\ 180°$.

Since the rules for construction of approximate root locus plots are well known, this program can be used to obtain the exact location of the roots in certain regions of the s-plane. The user selects a value of σ, say σ_1 and assumes a trial value for ω, say ω_1. The program then determines $G^{-1}(\sigma_1 + i\omega_1) = ke^{i\phi}$.

If ϕ is not within ±0.01 of 180°, a new trial value of $\omega = \omega_1$ is obtained. The equation for searching the correct value of is:

$$\omega = \omega_1\ (4 - \frac{\phi}{60}),\quad \text{where } \phi \text{ is in degrees.}$$

Note:

The search question is based on the assumption that ω_1 is greater than zero. This is not a limitation since the root locus is always symmetrical about the real axis of the s-plane. For display purposes, σ is represented as "a" by the 41C.

Example:

$$KG(s) = \frac{K(s + 1)(s + 2)}{s(s + 3)(s + 4)(s + 5)(s + 6)}$$

It is desired to obtain the complex roots for $\sigma = 0$, $\sigma = -1$, and $\sigma = -2.5$.

		initial ω
1.	$\sigma = 0$	$\omega = 5$
2.	$\sigma = -1$	$\omega = 6$
3.	$\sigma = -2.5$	$\omega = 4$

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 009	
[XEQ] [ALPHA] ROOTS 1 [ALPHA]	Z1?
1 [R/S]	Z2?
2 [R/S]	P1?
3 [R/S]	P2?
4 [R/S]	P3?
5 [R/S]	P4?
6 [R/S]	a?
0 [R/S]	W?
5 [R/S]	K=999.506
[R/S]	W=8.545
[R/S]	a?
1 [CHS] [R/S]	W?
6 [R/S]	K=519.846
[R/S]	W=6.824
[R/S]	a?
2.5 [CHS] [R/S]	W?
4 [R/S]	K=140.519
[R/S]	W=4.350

User Instructions

SIZE: 009

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program.			
2	Begin execution and input:		[XEQ] ROOTS 1	Z1?
	Z_1	Z_1	[R/S]	Z2?
	Z_2	Z_2	[R/S]	P1?
	P_1	P_1	[R/S]	P2?
	P_2	P_2	[R/S]	P3?
	P_3	P_3	[R/S]	P4?
	P_4	P_4	[R/S]	a?
3	σ	σ	[R/S]	W?
	initial guess for σ.	ω_1	[R/S]	
4	Solution will be displayed when			
	$179.99 < \phi < 180.01$.			K=
			[R/S]	W=
5	For a new root		[R/S]	a?
6	Go to step 3.			

Program Listings

Initialization

```
 01♦LBL "ROO
TS 1"
 02 "Z1?"
 03 PROMPT
 04 STO 01
 05 "Z2?"
 06 PROMPT
 07 STO 02
 08 "P1?"
 09 PROMPT
 10 STO 03
 11 "P2?"
 12 PROMPT
 13 STO 04
 14 "P3?"
 15 PROMPT
 16 STO 05
 17 "P4?"
 18 PROMPT
 19 STO 06
 20♦LBL A
 21 "a?"
 22 PROMPT
 23 "W?"
 24 PROMPT
 25♦LBL B
 26 STO 08
 27 X<>Y
 28 STO 07
 29 R-P
 30 RCL 06
 31 XEQ E
 32 RCL 05
 33 XEQ E
 34 RCL 04
 35 XEQ E
 36 RCL 03
 37 XEQ E
 38 RCL 02
 39 XEQ D
 40 RCL 01
 41 XEQ D
 42 X<>Y
 43 STO 00
 44 180
 45 -
 46 ABS
 47 .01
 48 X>Y?
 49 GTO C
 50 RCL 00
```

```
 51 60
 52 /
 53 4
 54 -
 55 CHS
 56 RCL 08
 57 *
 58 STO 08
 59 RCL 07
 60 RCL 08
 61 GTO B
 62♦LBL C
 63 RCL Z
 64 "K="
 65 XEQ d
 66 RCL 08
 67 "W="
 68 XEQ d
 69 GTO A
```

K

ω

Display

```
 70♦LBL d
 71 FIX 3
 72 ARCL X
 73 AVIEW
 74 STOP
 75 RTN
```

```
 76 STOP
 77♦LBL E
 78 RCL 07
 79 +
 80 RCL 08
 81 X<>Y
 82 R-P
 83 X<>Y
 84 RDN
 85 *
 86 RDN
 87 +
 88 R↑
 89 RTN
 90♦LBL D
 91 RCL 07
 92 +
 93 RCL 08
 94 X<>Y
 95 R-P
 96 X<>Y
 97 RDN
 98 /
 99 RDN
100 -
101 CHS
```

Program Listings

```
102 R↑
103 RTN
104 RCL 08
105 STOP
106 .END.
```

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	ϕ	50	
	Z1		
	Z2		
	P1		
	P2		
05	P3	55	
	P4		
	σ1		
	ω1		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 009 TOT. REG. 34 USER MODE
ENG ____ FIX 3 SCI ____ ON ____ OFF X
DEG X RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

AID TO ROOT LOCUS PLOTS I
– REAL POLES
PROGRAM REGISTERS NEEDED: 26

HEWLETT PACKARD
SOLUTION BOOK:
CONTROL SYSTEM

ROW 1 (1 – 2)

ROW 2 (2 – 8)

ROW 3 (8 – 14)

ROW 4 (15 – 21)

ROW 5 (22 – 31)

ROW 6 (31 – 37)

ROW 7 (37 – 44)

ROW 8 (44 – 51)

ROW 9 (52 – 62)

ROW 10 (62 – 67)

ROW 11 (68 – 73)

ROW 12 (74 – 85)

ROW 13 (86 – 97)

ROW 14 (98 – 106)

AID TO ROOT LOCUS PLOTS II - COMPLEX POLES

Given the forward transfer function of a unity-gain feedback system:

$$KG(s) = \frac{K(s + z_1)\ (s + z_2)}{s(s + p_1)\ (s + p_2)\ (s + \alpha + i\beta)\ (s + \alpha - i\beta)}$$

where $s = \sigma + i\omega$ is the complex frequency variable, and z_1, z_2, p_1, p_2, α and β are real numbers, the program helps in finding the roots of $1 + KS(s) = 0$, which determines the poles of the closed-loop system. It follows that at any point in the s-plane, which is a root of the above equation for some value of K, the argument of G(s) is equal to 180 degrees.

Since the rules for the construction of the approximate root locus plot are well known, this program can be used to obtain the exact location of the roots in critical regions of the s-plane. The user would select a value of σ, say σ_1 and assume a trial value for ω, say ω_1. The program then determines the modules M, and the argument, ϕ of $G^{-1}(s)$ at this point. If ϕ is not within the ±0.01 of 180° a new trial value of ω_2 is obtained using the equation $\omega_2 = \omega_1(4 - \phi/180)$.

The process is repeated until ϕ is within ±0.01 of 180°. The convergence may be slow when ω approaches zero.

Note:

The search equation is based on the assumption that ω is positive. This is not a limitation since the root-locus plot is always symmetrical about the real-axis of the s-plane. For display purposes, σ is represented as "a" by the 41C.

Example:

$$KG(s) = \frac{K}{s(s + 12 + 5i)\ (s + 12 - 5i)} = \frac{K(s + 1)\ (s + 2)}{s(s + 1)\ (s + 2)\ (s + 12 + 5i)\ (s + 12 - 5i)}$$

It is desired to obtain the exact locations of the roots for $\sigma = -1$, -2, and -4.

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 009	
[XEQ] [ALPHA] ROOTS 2 [ALPHA]	Z1?
1 [R/S]	Z2?
2 [R/S]	P1?
1 [R/S]	P2?
2 [R/S]	A?
12 [R/S]	B?
5 [R/S]	a?
1 [CHS] [R/S]	W?
12 [R/S]	K=2,749.323
[R/S]	W=11.134
[R/S]	a?
2 [CHS] [R/S]	W?
10 [R/S]	K=1,780.106
[R/S]	W=9.220
[R/S]	a?
4 [CHS] [R/S]	W?
5.5 [R/S]	K=656.209
[R/S]	W=5.002

User Instructions

SIZE: 009

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program.			
2	Begin execution and input:		[XEQ] ROOTS 2	Z1?
	Z_1	Z_1	[R/S]	Z2?
	Z_2	Z_2	[R/S]	P1?
	P_1	P_1	[R/S]	P2?
	P_2	P_2	[R/S]	A?
	α	α	[R/S]	B?
	β	β	[R/S]	a?
3	σ	σ	[R/S]	W?
	initial guess for ω.	ω_1	[R/S]	
4	Solution will be displayed when			
	$179.99 < \phi < 180.01$			K=
			[R/S]	W=
5	For a new root		[R/S]	a?
6	Go to step 3.			

Program Listings

```
 01♦LBL "ROO
TS 2"
 02 "Z1?"
 03 PROMPT          Initialization
 04 STO 01
 05 "Z2?"
 06 PROMPT
 07 STO 02
 08 "P1?"
 09 PROMPT
 10 STO 03
 11 "P2?"
 12 PROMPT
 13 STO 04
 14 "A?"
 15 PROMPT
 16 STO 05
 17 "B?"
 18 PROMPT
 19 STO 06
 20♦LBL A
 21 "a?"
 22 PROMPT
 23 "W?"
 24 PROMPT
                    ---------------
 25♦LBL B
 26 STO 08
 27 X<>Y
 28 STO 07
 29 RCL 05
 30 +
 31 X<>Y
 32 RCL 06
 33 +
 34 RCL 08
 35 RCL 06
 36 -
 37 R↑
 38 R-P
 39 RDN
 40 RDN
 41 XEQ E
 42 RCL 04
 43 RCL 07
 44 +
 45 RCL 08
 46 XEQ E
 47 RCL 03
 48 RCL 07
 49 +
 50 RCL 08
```

```
 51 XEQ E
 52 RCL 07
 53 RCL 08
 54 XEQ E
 55 RCL 02
 56 XEQ D
 57 RCL 01
 58 XEQ D
 59 X<>Y
 60 STO 00
 61 180
 62 -
 63 ABS
 64 .01
 65 X>Y?
 66 GTO C
 67 RCL 00
 68 60
 69 /              ω2
 70 4
 71 -
 72 CHS
 73 RCL 08
 74 *
 75 STO 08
 76 RCL 07
 77 RCL 08
 78 GTO B
 79♦LBL C
 80 FIX 3
 81 RCL Z          K
 82 "K="
 83 XEQ d
 84 RCL 08
 85 "W="          ω
 86 XEQ d
 87 GTO A
                  ---------------
 88♦LBL d
 89 ARCL X
 90 AVIEW          Display
 91 STOP
 92 RTN
                  ---------------
 93♦LBL D
 94 RCL 07
 95 +
 96 RCL 08
 97 X<>Y
 98 R-P
 99 X<>Y
100 RDN
101 /
```

Program Listings

```
102 RDN
103 -
104 CHS
105 R↑
106 RTN
107♦LBL E
108 X<>Y
109 R-P
110 X<>Y
111 RDN
112 *
113 RDN
114 +
115 R↑
116 RTN
117 .END.
```

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	φ	50	
	Z1		
	Z2		
	P1		
	P2		
05	α	55	
	β		
	σ1		
	ω1		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 009 TOT. REG. 35 USER MODE
ENG ____ FIX 3 SCI ____ ON ____ OFF X
DEG X RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
		NONE	

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

AID TO ROOT LOCUS PLOTS II
- COMPLEX POLES
PROGRAM REGISTERS NEEDED: 27

HEWLETT PACKARD
SOLUTION BOOK:
CONTROL SYSTEM

ROW 1 (1 - 2)

ROW 2 (2 - 8)

ROW 3 (8 - 15)

ROW 4 (16 - 23)

ROW 5 (23 - 33)

ROW 6 (34 - 44)

ROW 7 (45 - 53)

ROW 8 (54 - 60)

ROW 9 (61 - 67)

ROW 10 (68 - 78)

ROW 11 (78 - 83)

ROW 12 (84 - 89)

ROW 13 (89 - 100)

ROW 14 (101 - 112)

ROW 15 (113 - 117)

CLASSICAL CONTROL GAINS

This program computes the Ziegler-Nichols recommended settings for P, PI, and PID control. Data is required from one of two tests: the general control form is $G_c = k_c[1 + (1/T_i s) + T_D s]$.

A. <u>OPEN LOOP TEST</u> - Input step function, measure response, draw tangent at point of maximum response slope, scale L & R.



R = maximum slope
L = time intercept of maximum slope line

Then for P control, $k_c = 1/RL$; PI control, $k_c = .9/RL$, $T_i = 3.3L$; for PID control, $k_c = 1.2/RL$, $T_i = 2L$, $T_O = L/2$.

Note: Plant must be greater than first order for open loop test (s is Laplace operator).

B. <u>CLOSED LOOP TEST</u> - Increase k_c until plant is near instability (oscillating output). Let the magnitude of k_c at that point be called k_u, and the period of oscillation be called P_u.



Then for P control, let $k_c = .5\ k_u$

PI control, let $k_c = .45k_u$, $T_i = 0.83P_u$

PID control, let $k_c = 0.6k_u$, $T_i = .5P_u$, $T_O = P_{u/8}$

General control form is $G_c = k_c[1 + (1/T_i s) + T_O s]$

Note: Plant must be greater than second order for closed loop test.

Examples:

1. For the following open loop test data, compute the control coefficients for:

   a) P control
   b) PI control
   c) PID control

   (unit step input)



2. During a closed loop P control test, the plant output became oscillatory with k = k = 10, the period of oscillations was 50 sec. Compute coefficients for:

   a) P control
   b) PI control
   c) PID control

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 009	
[XEQ] [ALPHA] OPEN [ALPHA]	R?
0.5 [R/S]	L?
1 [R/S]	P
	KC=2.00
[R/S]	PI
	KC=1.80
[R/S]	TI=3.30
[R/S]	PID
	KC=2.40
[R/S]	TI=2.00
[R/S]	TO=0.50
[XEQ] [ALPHA] CLOSED [ALPHA]	KU?
10 [R/S]	PU?
50 [R/S]	P
	KC=5.00
[R/S]	PI
	KC=4.50
[R/S]	TI=41.50
[R/S]	PID
	KC=6.00
[R/S]	TI=25.00
[R/S]	TO=6.25

User Instructions

SIZE: 009

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program.			
2	For open loop test:		[XEQ] OPEN	R?
	input R	R	[R/S]	L?
	input L	L	[R/S]	P
				KC=
			[R/S]	PI
				KC=
			[R/S]	TI=
			[R/S]	PID
				KC=
			[R/S]	TI=
			[R/S]	TO=
3	For closed loop test:		[XEQ] CLOSED	KU?
	input Ku	Ku	[R/S]	PU?
	input Pu	Pu	[R/S]	P
				KC=
			[R/S]	PI
				KC=
			[R/S]	TI=
			[R/S]	PID
				KC=
			[R/S]	TI=
			[R/S]	TO=

Program Listings

```
 01♦LBL "OPE
N"
 02 "R?"
```

Open loop test

```
 03 PROMPT
 04 "L?"
 05 PROMPT
 06 STO 02
 07 *
 08 1/X
 09 STO 01
 10 STO 03
 11 .9
 12 *
 13 STO 04
 14 RCL 01
 15 1.2
 16 *
 17 STO 05
 18 RCL 02
 19 2
 20 *
 21 STO 07
 22 1.65
 23 *
 24 STO 06
 25 RCL 02
 26 2
 27 /
 28 STO 08
 29 CLX
 30 GTO C
 31♦LBL "CLO
SED"
```

Closed loop test

```
 32 "KU?"
 33 PROMPT
 34 "PU?"
 35 PROMPT
 36 STO 02
 37 RDN
 38 STO 01
 39 2
 40 /
 41 STO 03
 42 .9
 43 *
 44 STO 04
 45 RCL 01
 46 .6
 47 *
 48 STO 05
 49 RCL 02
 50 2
 51 /
 52 STO 07
 53 4
 54 /
 55 STO 08
 56 RCL 02
 57 .83
 58 *
 59 STO 06
 60 CLX
 61♦LBL C
 62 "  P"
```

Display

```
 63 AVIEW
 64 PSE
 65 RCL 03
 66 "KC="
 67 XEQ d
 68 "  PI"
 69 AVIEW
 70 PSE
 71 RCL 04
 72 "KC="
 73 XEQ d
 74 RCL 06
 75 "TI="
 76 XEQ d
 77 "  PID"
 78 AVIEW
 79 PSE
 80 RCL 05
 81 "KC="
 82 XEQ d
 83 RCL 07
 84 "TI="
 85 XEQ d
 86 RCL 08
 87 "T0="
 88♦LBL d
 89 ARCL X
 90 AVIEW
 91 STOP
 92 RTN
 93 .END.
```

00

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00		50	
	1/Rc or Kc		
	L or Pu		
	P-KC		
	PI-KC		
05	PID-KC	55	
	PI-Ti		
	PID-Ti		
	PID-To		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 009 TOT. REG. 34 USER MODE

ENG ___ FIX 2 SCI ___ ON ___ OFF X

DEG X RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

CLASSICAL CONTROL GAINS

PROGRAM REGISTERS NEEDED: 26

HEWLETT PACKARD
SOLUTION BOOK:
CONTROL SYSTEM

ROW 1 (1 - 4)

ROW 2 (4 - 14)

ROW 3 (15 - 22)

ROW 4 (23 - 31)

ROW 5 (31 - 34)

ROW 6 (34 - 43)

ROW 7 (44 - 55)

ROW 8 (56 - 62)

ROW 9 (63 - 68)

ROW 10 (68 - 74)

ROW 11 (75 - 77)

ROW 12 (78 - 84)

ROW 13 (84 - 89)

ROW 14 (89 - 93)

FIRST ORDER REGULATOR

Given a system: $\dot{x} = \frac{dx}{dt} = ax + bu$

where x = system state

u = system control

a, b constant.

This program solves the regulator problem, i.e., determines the optimal feedback gain to minimize the following performance index:

$$\text{performance index} = J = 1/2 \int_0^\infty (qx^2 + ru^2)\,dt$$

the solution is: $u = -cs$

where $c = \frac{b}{r}S$ and S is the positive solution to the Riccati equation:

$$0 = -2as + \frac{s^2b^2}{r} - q$$

Then: $\dot{x} = \bar{a}x$

where $\bar{a} = a - bc$

and $x = x_0 e^{-t/\tau}$

where $\tau = -\frac{1}{a}$

Note: $a \geqslant 0$, $r > 0$, and $b \neq 0$ must be satisfied to assure proper operation.

Example:

$$\dot{x} = ax + bu, \quad J = 1/2\int_0^\infty \left[\left(\frac{x}{x_{max}}\right)^2 + \left(\frac{u}{u_{max}}\right)^2\right] dt$$

1. $a = -1$, $b = 1$, $q = \left(\frac{1}{x_{max}}\right)^2 = 0$, $r = \left(\frac{1}{u_{max}}\right)^2 = 1$
2. $a = 1$, $b = 1$, $q = 0$, $r = 1$
3. $a = 1$, $b = 1$, $q = 1$, $r = 1$
4. $a = 1$, $b = 2$, $q = 3$, $r = 4$

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 005	
[XEQ] [ALPHA] REG 1 [ALPHA]	a?
1 [CHS] [R/S]	b?
1 [R/S]	Q?
0 [R/S]	R?
1 [R/S]	C=0.00
[R/S]	S=0.00
[R/S]	a=-1.00
[R/S]	T=1.00
[XEQ] [ALPHA] REG 1 [ALPHA]	a?
1 [R/S]	b?
1 [R/S]	Q?
0 [R/S]	R?
1 [R/S]	C=2.00
[R/S]	S=2.00
[R/S]	a=-1.00
[R/S]	T=1.00
[R/S]	Q?
1 [R/S]	R?
1 [R/S]	C=2.41
[R/S]	S=2.41
[R/S]	a=-1.41
[R/S]	T=0.71

User Instructions

SIZE: 005

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program.			
2	Begin execution and input:		[XEQ] REG 1	a?
	a	a	[R/S]	b?
	b	b	[R/S]	Q?
	q	q	[R/S]	R?
	r	r	[R/S]	
3	Solutions: c			C=
	s		[R/S]	S=
	$\bar{a}$		[R/S]	a=
	τ		[R/S]	T=
4	To change only q and r:		[R/S]	Q?
	input q	q	[R/S]	R?
	input r	r	[R/S]	
	go to step 3.			

Program Listings

```
01♦LBL "REG
1"
02 "a?"
03 PROMPT          Initialization
04 STO 01
05 "b?"
06 PROMPT
07 STO 02
08♦LBL A
09 "Q?"
10 PROMPT
11 STO 03
12 "R?"
13 PROMPT
14 STO 04
15 RCL 01
16 RCL 02
17 /
18 X↑2
19 RCL 03
20 RCL 04
21 /
22 +
23 SQRT
24 RCL 01
25 RCL 02
26 /
27 +
28 "C="            C
29 XEQ d
30 RCL 04
31 *
32 RCL 02
33 /
34 "S="            S
35 XEQ d
36 RCL 01
37 X↑2
38 RCL 02
39 X↑2
40 RCL 03
41 *
42 RCL 04
43 /
44 +
45 SQRT
46 CHS
47 "a="            a
48 XEQ d
49 1/X
50 CHS
51 "T="
52 XEQ d
53 GTO A
54♦LBL d
55 ARCL X
56 AVIEW
57 STOP
58 RTN
59 .END.
```

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00		50	
	a		
	b		
	q		
	r		
05		55	
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 005 TOT. REG. 18 USER MODE
ENG ____ FIX 2 SCI ____ ON ____ OFF X
DEG X RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

FIRST ORDER REGULATOR

HEWLETT PACKARD
SOLUTION BOOK:
CONTROL SYSTEM

PROGRAM REGISTERS NEEDED: 14

ROW 1 (1 - 3)

ROW 2 (4 - 11)

ROW 3 (12 - 22)

ROW 4 (23 - 31)

ROW 5 (32 - 40)

ROW 6 (41 - 49)

ROW 7 (50 - 55)

ROW 8 (55 - 59)

SECOND ORDER REGULATOR

Given a system and a quadratic performance index as follows:

$$\begin{bmatrix} \dot{x}_1 \\ \dot{x}_2 \end{bmatrix} = \begin{bmatrix} a_{11} & a_{12} \\ a_{21} & a_{22} \end{bmatrix} \begin{bmatrix} x_1 \\ x_2 \end{bmatrix} = \begin{bmatrix} 0 \\ b \end{bmatrix} u$$

$$J = 1/2 \int_0^\infty \left\{ [x_1 \quad x_2] \begin{bmatrix} q_{11} & q_{12} \\ q_{21} & q_{22} \end{bmatrix} \begin{bmatrix} x_1 \\ x_2 \end{bmatrix} + u^2 r \right\} dt$$

where: x_1, x_2 are the system states

u is the control

J is the performance index to be minimized.

The optimal control is given by:

$$u = -c_1,\ x_1 - c_2 x_2 \quad \text{and} \quad [\dot{x}] = [A][x] \qquad [A] = [A] - \begin{bmatrix} o \\ b \end{bmatrix} [c_1 - c_2]$$

where:

$$c_1 = \frac{1}{b} (c_1 * a_{11} c_2 *) \qquad c_2 = \frac{1}{b} (a_{12} c_2 *)$$

$$c_1 * = a_1 + \sqrt{a_1^2 + q_1 \frac{b^2}{r}} \qquad c_2 * = a_2 + \sqrt{a_2^2 + 2c_1 * + q_3 \frac{b^2}{r}}$$

$$a_1 = a_{12} a_{21} - a_{11} a_{22} \qquad q_1 = q_{11} - 2 \frac{q_{12} a_{11}}{a_{12}} + \frac{q_{22} a_{11}^2}{a_{12}^2}$$

$$a_2 = a_{11} + a_{22} \qquad q_3 = q_{22}$$

$$q_{11},\ q_{12},\ q_{12} \geqslant 0; \quad q_{11} + q_{12} + q_{13} = 0$$

Example:

$$\ddot{y} + \theta\dot{y} + \phi = u \qquad \text{second order system}$$

Let $x_1 = y$, $x_2 = \dot{y}$

$$\begin{bmatrix} \dot{x}_1 \\ \dot{x}_2 \end{bmatrix} = \begin{bmatrix} 0 & 1 \\ -\phi & -\theta \end{bmatrix} \begin{bmatrix} x_1 \\ x_2 \end{bmatrix} + \begin{bmatrix} 0 \\ 1 \end{bmatrix} u \qquad \text{suppose } \phi = 2,\ \theta = 3$$

Find the optimal control that minimizes:

1. $J = 1/2 \int_0^\infty (x_1^2 + x_2^2 + u^2)\,dt$, i.e., $q_{11} = 1$, $q_{12} = 0$, $q_{22} = 1$, $r = 1$.
2. $J = 1/2 \int_0^\infty (4x_1^2 + u^2)\,dt$, i.e., $q_{11} = 4$, $q_{12} = 0$, $q_{22} = 0$, $r = 1$.

$$A = \begin{bmatrix} 0 & 1 \\ -2 & -3 \end{bmatrix} \qquad b = 1,\ r = 1$$

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 010	
[XEQ] [ALPHA] REG 2 [ALPHA]	A11?
0 [R/S]	A12?
1 [R/S]	A21?
2 [CHS] [R/S]	A22?
3 [CHS] [R/S]	b?
1 [R/S]	Q11?
1 [R/S]	Q12?
0 [R/S]	Q22?
1 [R/S]	R?
1 [R/S]	C1=0.236
[R/S]	C2=0.236
[R/S]	Q11?
4 [R/S]	Q12?
0 [R/S]	Q22?
0 [R/S]	R?
1 [R/S]	C1=0.828
[R/S]	C2=0.264

User Instructions

SIZE: 010

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program.			
2	Begin execution:		[XEQ] REG 2	A11?
	input A matrix	a11	[R/S]	A12?
		a12	[R/S]	A21?
		a21	[R/S]	A22?
		a22	[R/S]	b?
	input b.	b	[R/S]	Q11?
3	Input Q	Q11	[R/S]	Q12?
		Q12	[R/S]	Q22?
		Q22	[R/S]	R?
	input r.	r	[R/S]	C1=
			[R/S]	C2=
4	For a change of Q and R:		[R/S]	Q11?
5	Go to step 3.			
6	Closed loop dynamics:		[XEQ] A	A11=
	matrix [A]		[R/S]	A12=
			[R/S]	A21=
			[R/S]	A22=

Program Listings

Initialization

```
01♦LBL "REG·
2"
02 "A11?"
03 PROMPT
04 STO 01
05 "A12?"
06 PROMPT
07 STO 02
08 "A21?"
09 PROMPT
10 STO 03
11 "A22?"
12 PROMPT
13 STO 04
14 "b"
15 PROMPT
16 STO 05
17♦LBL C
18 "Q11?"
19 PROMPT
20 "Q12?"
21 PROMPT
22 "Q22?"
23 PROMPT
24 "R?"
25 PROMPT
```

```
26 STO 07
27 /
28 STO 08
29 LASTX
30 *
31 RCL 01
32 *
33 RCL 02
34 /
35 X<>Y
36 2
37 *
38 -
39 RCL 01
40 *
41 RCL 02
42 /
43 +
44 RCL 07
45 /
46 STO 07
47 RCL 02
48 RCL 03
49 *
50 RCL 01
51 RCL 04
52 *
53 -
54 STO 06
55 RCL 07
56 SQRT
57 RCL 05
58 *
59 RCL 06
60 R-P
61 RCL 06
62 +
63 STO 06
64 2
65 *
66 RCL 05
67 X↑2
68 RCL 08
69 *
70 +
71 SQRT
72 RCL 01
73 RCL 04
74 +
75 STO 07
76 R-P
77 RCL 07
78 +
79 STO 07
80 RCL 01
81 *
82 RCL 06
83 +
84 STO 06
85 RCL 05
86 /
```

C1

```
87 "C1="
88 XEQ d
89 RCL 07
90 RCL 02
91 *
92 STO 07
93 RCL 05
94 /
```

C2

```
95 "C2="
96 XEQ d
97 GTO C
```

[A]

```
98♦LBL A
99 RCL 02
100 RCL 03
101 RCL 06
```

Program Listings

```
102 -
103 RCL 04
104 RCL 07
105 -
106 RCL 01
107 "A11="
108 XEQ d
109 R↑
110 "A12="
111 XEQ d
112 R↑
113 "A21="
114 XEQ d
115 R↑
116 "A22="
117♦LBL d
118 FIX 3
119 ARCL X
120 AVIEW
121 STOP
122 RTN
123 .END.
```

Display

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00		50	
	a11		
	a12		
	a21		
	a22		
05	b	55	
	temp		
	temp		
	temp		
	R→P		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 010 TOT. REG. 38 USER MODE

ENG ____ FIX 3 SCI ____ ON ____ OFF X

DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

SECOND ORDER REGULATOR

PROGRAM REGISTERS NEEDED: 29

HEWLETT PACKARD
SOLUTION BOOK:
CONTROL SYSTEM

ROW 1 (1 - 2)

ROW 2 (2 - 8)

ROW 3 (8 - 14)

ROW 4 (15 - 20)

ROW 5 (20 - 25)

ROW 6 (26 - 38)

ROW 7 (39 - 51)

ROW 8 (52 - 64)

ROW 9 (65 - 77)

ROW 10 (78 - 87)

ROW 11 (88 - 95)

ROW 12 (96 - 103)

ROW 13 (104 - 110)

ROW 14 (110 - 113)

ROW 15 (114 - 118)

ROW 16 (119 - 123)

NOTES

NOTES

Hewlett-Packard Software

In terms of power and flexibility, the problem-solving potential of the HP-41 programmable calculator is nearly limitless. And in order to see the practical side of this potential, HP has different types of software to help save you time and programming effort. Every one of our software solutions has been carefully selected to effectively increase your problem-solving potential. Chances are, we already have the solutions you're looking for.

Application Pacs

To increase the versatility of your HP-41, HP has an extensive library of "Application Pacs". These programs transform your HP-41 into a specialized calculator in seconds. Included in these pacs are detailed manuals with examples, miniature plug-in Application Modules, and keyboard overlays. Every Application Pac has been designed to extend the capabilities of the HP-41.

You can choose from:

Aviation (Pre-Flight Only) 00041-15018
Clinical Lab 00041-15024
Circuit Analysis 00041-15024
Financial Decisions 00041-15004
Mathematics 00041-15003
Structural Analysis 00041-15021
Surveying 00041-15005
Securities 00041-15026
Statistics 00041-15002
Stress Analysis 00041-15027
Games 00041-15022
Home Management 00041-15023
Machine Design 00041-15020
Navigation 00041-15017
Real Estate 00041-15016
Thermal and Transport Science 00041-15019
Petroleum Fluids 00041-15039

Users' Library

The Users' Library provides the best programs from contributors and makes them available to you. By subscribing to the HP-41 Users' Library you'll have at your fingertips literally hundreds of different programs from many different application areas.

*Users' Library Solutions Books

Hewlett-Packard offers a wide selection of Solutions Books complete with user instructions, examples, and listings. These solution books will complement our other software offerings and provide you with a valuable tool for program solutions.

You can choose from:

Business Stat/Marketing/Sales 00041-90094
Home Construction Estimating 00041-90096
Lending, Saving and Leasing 00041-90086
Real Estate 00041-90136
Small Business 00041-90137
Geometry 00041-90084
High-Level Math 00041-90083
Test Statistics 00041-90082
Antennas 00041-90093
Chemical Engineering 00041-90100
Control Systems 00041-90092
Electrical Engineering 00041-90088
Fluid Dynamics and Hydraulics 00041-90139
Games II 00041-90443
Civil Engineering 00041-90089
Heating, Ventilating & Air Conditioning 00041-90140
Mechanical Engineering 00041-90090
Solar Engineering 00041-90138
Calendars 00041-90145
Cardiac/Pulmonary 00041-90097
Chemistry 00041-90102
Games 00041-90099
Optometry I (General) 00041-90143
Optometry II (Contact Lens) 00041-90144
Physics 00041-90142
Surveying 00041-90141
Time Module Solutions 00041-90395

*Some books require additional memory modules to accomodate all programs.

CONTROL SYSTEMS

FREQUENCY RESPONSE OF A TRANSFER FUNCTION
BODE OF TRANSFER FUNCTION THAT HAS EACH POLE AND ZERO GIVEN
BODE OF THIRD-ORDER OVER FOURTH-ORDER TRANSFER FUNCTION
BODE OF THIRD-ORDER OVER THIRD-ORDER TIMES S^N TRANSFER FUNCTION
ROUTH TEST FOR CONTINUOUS AND DISCRETE TIME SYSTEM STABILITY
CONVERT FREQUENCY RESPONSE —— OPEN LOOP, CLOSED LOOP
AID TO ROOT LOCUS PLOTS I —— REAL POLES
AID TO ROOT LOCUS PLOTS II —— COMPLEX POLES
CLASSICAL CONTROL GAINS
FIRST ORDER REGULATOR
SECOND ORDER REGULATOR

HEWLETT PACKARD

00041-90092 Rev. B 10/83 Printed in U.S.A.